VIII Межвузовская олимпиада по информационным технологиям, Липецк, 2019

Задания для олимпиады по информационным технологиям
Решить предложенные задачи средствами Microsoft Excel. Сохранять решение задач на разных листах одной книги. Имена листам присваивать по номеру задачи: Задание1, Задание2, Задание3.
Задача 1 (3 балла)
Подготовить шаблон ведомости промежуточной аттестации для группы до 30 человек. В ведомость проставляется оценка по предмету. Студент считается аттестованным если он набрал по предмету 7 или более баллов 

После окончания аттестации необходимо подсчитать количество не аттестованных студентов по предметам и количество человек с указанным числом «заваленных» предметов (от 0 до 5). Также требуется выделить низкие оценки (<7).
Исходные данные представлены в файле.

Задача 2 (4 балла)
Вася, Петя и Миша решили провести соревнования по бадминтону и попросили вас подготовить таблицу соревнований.
Так как вам не хочется постоянно заносить результаты игр в таблицу, то вы решили автоматизировать этот процесс подготовив автоматический шаблон.
Поля, заполненные бледно-розовым цветом, заполняются спортсменами, а все остальные значения вычисляются по формулам.
Игры проводятся до 2-х побед, а в партии выигрывает тот, кто первым наберёт 21 очко. За победу спортсмен получает 1 очко. Результаты игр заносятся в нижнюю часть таблицы. Для удобства контроля выигравший спортсмен выделяется цветом и полужирным шрифтом.

По результатам игры требуется подсчитать общий счёт по партиям (верхняя часть ячейки таблицы) и результат игры.
[image: image2.png]TypHup apyseit

Ne KomaHga 1 2 3 Ouku |Maptum|Mecto
0:2
Baca 0 0:2
1 0
Met 0 21 2 | 4
2 1 1
[ 12
Mwuwa 0 1:2
3 0
CYéT No napTuam
1 MNeta Mwwa 21:15 | 9:21 |21:19 2:1
2 Baca Meta 7:21 |12:21 0:2
3 Muuwa Bacsa





Исходные данные представлены в файле. Пример форматирования представлен выше.
Задача 3 (3 балла)

В технологическом техникуме заболел кассир и подготовить ведомость на выплату зарплаты 100 сотрудникам попросили одного из работников. К сожалению, рабочий привык к английской раскладке и заменил русские буквы на сходные по начертанию* латинские, оставив остальные буквы как есть. И эта проблема была обнаружена только после сортировки работников по фамилиям.

Вам требуется восстановить русские буквы в фамилиях сотрудников и отсортировать таблицу в правильном порядке.
В этой задаче достаточно получить правильный результат.
Исходные данные представлены в файле.

* Для справки

Замене подверглись следующие буквы: А, В, Е, К, М, Н, О, Р, С, Т, У, Х.
Разбор Задачи 1

Для решения задачи следует посчитать в каждой строке количество оценок, которые меньше 7. Для этого в ячейку, например, K4 можно добавить формулу: =СЧЁТЕСЛИ(C4:G4;"<7"). Если получилось положительное число, то студент не имеет аттестацию хотя бы по одному предмету. Осталось проверить это. В ячейку H4 запишем формулу: =ЕСЛИ(K4>0;"н/атт";"атт"). Останется только скрыть столбец K.
Можно совместить эти формулы в одной ячейке и получить формулу: =ЕСЛИ(СЧЁТЕСЛИ(C4:G4;"<7")>0;"н/атт";"атт").

Ещё вариант: =ЕСЛИ(МИН(C4:G4)<7;"н/атт";"атт").

Полученные формулы следует растянуть на всю таблицу.

Для подсчёта неаттестаций по предмету подсчитаем количество оценок меньших 7 по каждому столбцу по формуле: =СЧЁТЕСЛИ(C4:C33;"<7"), а затем распространим её на все 5 столбцов.

Для подсчёта количества аттестованных по всем предметам студентов придётся использовать дополнительный столбец с количеством неаттестаций у каждого студента, что было рассмотрено выше. Теперь потребуется подсчитать в этом столбце количество 0: =СЧЁТЕСЛИ(K4:K33;0). Подобную формулу надо написать и в остальных 5 ячейках. Можно использовать абсолютную ссылку для данных и относительную для количества. Полученную формулу: =СЧЁТЕСЛИ($K$4:$K$33;C38) растянем на все 6 позиций и получим требуемый результат.
Для выделения низких оценок используем условное форматирование.

Разбор Задачи 2
Решение задачи лучше разбить на несколько этапов.

1. Для каждой партии (строки 13-15) следует подсчитать количество воланов, забитых каждым игроком в партии.

2. Для определения счёта в партии подсчитаем сколько раз первый игрок набрал очков больше, чем второй.

3. Соберём полученные результаты в нужном порядке в ячейках.

4. Подсчитаем итоговые результаты.

5. Скроем промежуточные ячейки (столбцы).

Используем для вычисления промежуточных результатов столбцы, начиная с L.
1. Счёт в партии разделяется символом “:”. Левая часть относится к первому спортсмену, а правая ко второму. Для поиска позиции этого символа используем функцию ПОИСК, которая возвращает позицию первого вхождения символа в текст. Для проверки заполненности ячейки проверим длину текста в ней (функция ДЛСТР). Для вырезания первых k символов используем функцию ЛЕВСИМВ(текст, k). Полученную строку следует преобразовать в число (функция ЗНАЧЕН), чтобы можно было использовать в последующих расчётах.
Формула для подсчёта числа воланов, забитых первым игроком в первой партии (ячейка E13), примет вид: =ЗНАЧЕН(ЛЕВСИМВ(E13;ПОИСК(":";E13)-1)). Завернув в условие получим: =ЕСЛИ(И(ДЛСТР(E13)>2;ПОИСК(":";E13)>1);ЗНАЧЕН(ЛЕВСИМВ(E13;ПОИСК(":";E13)-1));0).
Для подсчёта числа воланов, забитых вторым игроком в первой партии, заменим в формуле функцию ЛЕВСИМВ, на ПСТР, которая возвращает среднюю часть текста. Формула примет вид: =ЕСЛИ(И(ДЛСТР(E13)>2;ПОИСК(":";E13)>1);ЗНАЧЕН(ПСТР(E13;ПОИСК(":";E13)+1;2));0).

Аналогично подсчитаем воланы и в двух других партиях.

Можно скопировать формулы в две строки и перетянуть формулы. 😊
[image: image5.png]- Fe | =ECAM(M(A/NICTP(E13)>2;MOMCK(":";E13)>1);3HAYEH(IEBCUIMB(E13;TIOMCK("";E13)-1));0)
T T T T T T

E F G H I J K L MNO P Q

5 21:15]9:221 [21:19 21 12 2 1 21| 15 0| 21/ 21| 10




2. Для подсчёта счёта в партиях сложим сколько раз счёт в партии у первого игрока был больше, чем у второго. Получим формулу для первого игрока: =ЕСЛИ(L13>M13;1;0)+ЕСЛИ(N13>O13;1;0)+ЕСЛИ(P13>Q13;1;0).

Для второго используем такую же формулу, поменяв местами аргументы в ЕСЛИ: =ЕСЛИ(M13>L13;1;0)+ЕСЛИ(O13>N13;1;0)+ЕСЛИ(Q13>P13;1;0).
Сцепив полученные результаты с помощью функции СЦЕПИТЬ, получим формулу: =ЕСЛИ(J13+K13>0;СЦЕПИТЬ(J13;":";K13);"").
Сцепив результаты в другом порядке, получим обратный счёт по партиям.

3. Полученные результаты продублируем в исходной таблице в следующем порядке.

Представленные формулы обёрнуты в функцию ЕСЛИ, чтобы сделать вывод результатов незаполненной таблицы более привычным.

Для подсчёта результата партии сравним левую и правую половину счёта, как мы уже делали ранее. Например, для ячейки D5 получим формулу:

=ЕСЛИ(ДЛСТР(D4)<3;"";ЕСЛИ(ЛЕВСИМВ(D4;1)>ПРАВСИМВ(D4;1);1;0)).

4. Очки – это сумма по нижней строке: =СУММ(C5:E5).
Партии – соотношение выигранных и проигранных партий: =СЦЕПИТЬ(J14+K15;":";K14+J15).

Место проставляется вручную, т.к. при равном количество очков есть разные варианты определения победителя.

Для удобства добавим выделение победителя в каждой партии используя условное форматирование.

5. Скроем промежуточные столбцы с I до Q.
Разбор Задачи 3
Задача предполагает 2 решения: с помощью функций Excel и выполнение команд в ручном режиме. Рассмотрим оба варианта, начиная с первого.
Решение можно разбить на следующие этапы:

1. Подготовим таблицу замены букв.
2. Разбить строку с фамилией на отдельные буквы.

3. Заменить необходимые буквы.

4. Собрать фамилию обратно.

1. Подготовим таблицу замены букв. Можно дать имя диапазону, например, ТаблицаЗамены. В первом столбце укажем латинскую букву, а во втором – соответствующую русскую букву для замены.

2. Скопируем фамилию в отдельную ячейку правее таблицы и подсчитаем число символов в строке с помощью функции ДЛСТР(G4). Найдём максимальную длину фамилии с помощью функции МАКС(F4:F103). Чуть выше добавим номера букв, от 1 до 17. Теперь вырежем отдельную букву фамилии, указывая в качестве текста ячейку с фамилией, начальную позицию возьмём из строки выше, и зададим вырезку одного символа. Формула примет вид: ПСТР($G4;H$3;1).
Распространим формулу по строке на все буквы фамилии, а затем всю строку протянем вниз на все 100 фамилий. Чтобы выполнить данную операцию следует при указании ссылок на ячейки использовать смешанный стиль ссылок, зафиксировав для фамилии имя столбца, а для номера буквы – номер столбца.

3. Для замены используем функцию ВПР, которую сразу обернём в функцию ЕСЛИ. Если буква есть в таблице замены, то возвращаем новый вариант, оставив исходную букву в противном случае. Следующую формулу расположим правее букв фамилии:
 =ЕСЛИ(ЕНД(ВПР(H4;$B$106:$C$129;2;0));H4;ВПР(H4;$B$106:$C$129;2;0)).
4. Для обратной сборки фамилии используем функцию СЦЕПИТЬ:

СЦЕПИТЬ(AC4;AD4;AE4;AF4;AG4;AH4;AI4;AJ4;AK4;AL4;AM4;AN4;AO4;AP4;AQ4;AR4;AS4;AT4;AU4;AV4).

Полученную фамилию можно использовать в качестве ключа для сортировки исходного текста с фамилиями, а можно заменить ими исходные фамилии и удалить все произведённые вычисления. Для этого следует выделить фамилии, скопировать и вставить только значения.
Второй вариант позволяет выполнить все операции заметно быстрее. 😊
Можно воспользоваться встроенной командой “Найти и заменить”, которую используем 12 раз (по числу заменяемых букв). Заменим все вхождения и радостно представим результат работы.

